Lecture 15: Surface Integrals and Some Related Theorems

3.016

Reading:
Kreyszig Sections: 10.4, 10.5, 10.6, 10.7 (pages439–444, 445–448, 449–458, 459–462)

Green's Theorem for Area in Plane Relating to its Bounding Curve

Reappraise the simplest integration operation, $g(x) = \int f(x)dx$. Temporarily ignore all the tedious mechanical rules of finding and integral and concentrate on what integration *does*.

Integration replaces a fairly complex process—adding up all the contributions of a function $f(x)$—with a clever new function $g(x)$ that only needs end-points to return the result of a complicated summation.

It is perhaps initially astonishing that this complex operation on the interior of the integration domain can be incorporated merely by the domain's endpoints. However, careful reflection provides a counterpoint to this marvel. How could it be otherwise? The function $f(x)$ is specified and there are no surprises lurking along the x-axis that will trip up dx as it marches merrily along between the endpoints. All the facts are laid out and they willingly submit to the process their preordination by $g(x)$ by virtue of the endpoints.[7]

The idea naturally translates to higher dimensional integrals and these are the basis for Green's theorem in the plane, Stoke's theorem, and Gauss (divergence) theorem. Here is the idea:

[7]I do hope you are amused by the evangelistic tone. I am a bit punchy from working non-stop on these lectures and wondering if anyone is really reading these notes. Sigh.



Figure 15-11: An irregular region on a plane surrounded by a closed curve. Once the closed curve (the edge of region) is specified, the area inside it is already determined. This is the simplest case as the area is the integral of the function $f = 1$ over $dxdy$. If some other function, $f(x, y)$, were specified on the plane, then its integral is also determined by summing the contributions along the boundary. This is a generalization $g(x) = \int f(x)dx$ and the basis behind Green's theorem in the plane.

The analog of the "*Fundamental Theorem of Differential and Integral Calculus*"[8] for a region $\mathcal{R}$ bounded in a plane with normal $\hat{k}$ that is bounded by a curve $\partial\mathcal{R}$ is:

$$\int\int_{\mathcal{R}} (\nabla \times \vec{F}) \cdot \hat{k} dxdy = \oint_{\partial\mathcal{R}} \vec{F} \cdot d\vec{r} \qquad (15\text{-}1)$$

The following figure motivates Green's theorem in the plane:

[8] This is the theorem that implies the integral of a derivative of a function is the function itself (up to a constant).

3.016
3.016 Home
Full Screen
Close
Quit



Figure 15-12: Illustration of how a vector valued function in a planar domain "spills out" of domain by evaluating the curl everywhere in the domain. Within the domain, the rotational flow ($\nabla \times F$) from one cell moves into its neighbors; however, at the edges the local rotation is a net loss or gain. The local net loss or gain is $\vec{F} \cdot (dx, dy)$.

The generalization of this idea to a surface $\partial\mathcal{B}$ bounding a domain $\mathcal{B}$ results in Stokes' theorem, which will be discussed later.

In the following example, Green's theorem in the plane is used to simplify the integration to find the potential above a triangular path that was evaluated in a previous example. The result will be a considerable increase of efficiency of the numerical integration because the two-dimensional area integral over the interior of a triangle is reduced to a path integral over its sides.

The objective is to turn the integral for the potential

$$E(x, y, z) = \int\int_R \frac{d\xi d\eta}{\sqrt{(x-\xi)^2 + (y-\eta)^2 + z^2}} \tag{15-2}$$

into a path integral using Green's theorem in the x–y plane:

$$\int\int_R \left(\frac{\partial F_2}{\partial x} - \frac{\partial F_1}{\partial y} \right) dxdy = \int_{\partial R} (F_1 dx + F_2 dy) \tag{15-3}$$

3.016 Home
Full Screen
Close
Quit

To find the vector function $\vec{F} = (F_1, F_2)$ which matches the integral in question, set $F_2 = 0$ and integrate to find F_1 via

$$\int \frac{d\eta}{\sqrt{(x-\xi)^2 + (y-\eta)^2 + z^2}} \tag{15-4}$$

Lecture 15 MATHEMATICA® Example 1

Turning an integral over a domain into an integral over its boundary

notebook (non-evaluated) pdf (evaluated) html (evaluated)

Here we turn the two dimensional numerical integration which requires $\mathcal{O}(N^2)$ calculations into an integration around the boundary which requires $\mathcal{O}(N)$ evaluations for the same accuracy. The path of integration must be determined (i.e., $(x(t), y(t))$) and then the integration is obtained via $(dx, dy) = (x'dt, y'dt)$.

1: Finding F_1 as indicated above is obtained easily with `Integrate`.

2: The bottom part of the triangle can be written as the curve: $(\zeta(t), \eta(t)) = (t - \frac{1}{2}, 0)$ for $0 < t < 1$; the integrand over that side is obtained by suitable replacement.

5: The remaining two legs of the triangle can be written similarly as: $((1-t)/2, \sqrt{3}t/2)$ and $(-t/2, \sqrt{3}(1-t)/2)$.

6: This is the integrand for the entire triangle to be integrated over $0 < t < 1$.

7: There is no free lunch—the closed form of the integral is either unknown or takes too long to compute.

8: However, `NIntegrate` is much more efficient because the problem has been reduced to a single integral instead of the double integral in the previous example.



UseGreen's theorem to replace the area integral with path integral

$$\int\!\!\int_{\triangle}\left(\frac{\partial F_2}{\partial x} - \frac{\partial F_1}{\partial y}\right) dx\, dy = \int_{\triangle} F_1\, dx + F_2\, dy$$

Try to find a vector function (F_1, F_2) that matches the integrand over the domain. Let $F_2 = 0$, then to find F_1 which when differentiated by η gives $1/r$

```
1  F1[x_, y_, z_] = -Integrate[1/Sqrt[(x - ξ)^2 + (y - η)^2 + z^2], η]

2  Bottomside = F1[x, y, z] /. {ξ → t - 1/2, η → 0} // Simplify

3  BottomContribution = Integrate[Bottomside, {t, 0, 1},
      Assumptions → x ∈ Reals && y ∈ Reals && z > 0]

4  NEside = F1[x, y, z] /. {ξ → (1 - t)/2, η → Sqrt[3] t/2} // Simplify

5  NWside = F1[x, y, z] /. {ξ → -t/2, η → Sqrt[3] (1 - t)/2} // Simplify
```

$$\int_{\triangle} F_1\, dx = \int_0^1 \text{NEside}\,\frac{-dt}{2} + \int_0^1 \text{NWside}\,\frac{-dt}{2} + \int_0^1 \text{Bottomside}\, dt$$

```
6  integrand = Simplify[-(NEside + NWside)/2 + Bottomside]

7  PotXYZ = Integrate[integrand, {t, 0, 1},
      Assumptions → x ∈ Reals && y ∈ Reals && z > 0]

8  Pot[X_, Y_, Z_] := NIntegrate[
      Evaluate[integrand /. {x → X, y → Y, z → Z}], {t, 0, 1}]

9  ContourPlot[Pot[a, b, 1/10], {a, -1, 1}, {b, -.5, 1.5},
     Contours → 8, ColorFunction → (Hue[1 - 0.66*#] &)]
```

Representations of Surfaces

Integration over the plane $z = 0$ in the form of $\int f(x, y)dxdy$ introduces surface integration—over a planar surface—as a straightforward extension to integration along a line. Just as integration over a line was generalized to integration over a curve by introducing two or three variables that depend on a *single* variable (e.g., $(x(t), y(t), z(t))$), a surface integral can be conceived as introducing three (or more) variables that depend on two parameters (i.e., $(x(u, v), y(u, v), z(u, v))$).

However, there are different ways to formulate representations of surfaces:

Surfaces and interfaces play fundamental roles in materials science and engineering. Unfortunately, the mathematics of surfaces and interfaces frequently presents a hurdle to materials scientists and engineering. The concepts in surface analysis can be mastered with a little effort, but there is no escaping the fact that the algebra is tedious and the resulting equations are onerous. Symbolic algebra and numerical analysis of surface alleviates much of the burden.

Most of the practical concepts derive from a second-order Taylor expansion of a surface near a point. The first-order terms define a tangent plane; the tangent plane determines the surface normal. The second-order terms in the Taylor expansion form a matrix and a quadratic form that can be used to formulate an expression for curvature. The eigenvalues of the second-order matrix are of fundamental importance.

The Taylor expansion about a particular point on the surface takes a particularly simple form if the origin of the coordinate system is located at the point and the z-axis is taken along the surface normal as illustrated in the following figure.



Figure 15-13: **Parabolic approximation to a surface and local eigenframe.** The surface on the left is a second-order approximation of a surface at the point where the coordinate axes are drawn. The surface has a local normal at that point which is related to the cross product of the two tangents of the coordinate curves that cross at the that point. The three directions define a coordinate system. The coordinate system can be translated so that the origin lies at the point where the surface is expanded and rotated so that the normal $\hat{n}$ coincides with the z-axis as in the right hand curve.

In this coordinate system, the Taylor expansion of $z = f(x, y)$ must be of the form

$$\Delta z = 0dx + 0dy + \frac{1}{2}(dx, dy) \begin{pmatrix} \frac{\partial^2 f}{\partial x^2} & \frac{\partial^2 f}{\partial x \partial y} \\ \frac{\partial^2 f}{\partial x \partial y} & \frac{\partial^2 f}{\partial y^2} \end{pmatrix} \begin{pmatrix} dx \\ dy \end{pmatrix}$$

If this coordinate system is rotated about the z-axis into its eigenframe where the off-diagonal components vanish, then the two eigenvalues represent the maximum and minimum curvatures. The sum of the eigenvalues is invariant to transformations and the sum is known as the *mean curvature* of the surface. The product of the eigenvalues is also invariant—this quantity is known as the *Gaussian curvature*.

The method in the figure suggests a method to calculate the normals and curvatures for a surface. Those results are tabulated below.

Level Set Surfaces: Tangent Plane, Surface Normal, and Curvature

$$F(x, y, z) = \text{const}$$

Tangent Plane ($\vec{x} = (x, y, z)$, $\vec{\xi} = (\xi, \eta, \zeta)$)

$$\nabla F \cdot (\vec{\xi} - \vec{x}) \text{ or } \frac{\partial F}{\partial x}(\xi - x) + \frac{\partial F}{\partial y}(\eta - y) + \frac{\partial F}{\partial z}(\zeta - z)$$

Normal

$$\frac{\xi - x}{\frac{\partial F}{\partial x}} = \frac{\eta - y}{\frac{\partial F}{\partial y}} = \frac{\zeta - z}{\frac{\partial F}{\partial z}}$$

Mean Curvature

$$\nabla \cdot \left(\frac{\nabla F}{\|\nabla F\|} \right) \text{ or}$$

$$\frac{\left[\begin{array}{c} \left(\frac{\partial^2 F}{\partial y^2} + \frac{\partial^2 F}{\partial z^2} \right) (\frac{\partial F}{\partial x})^2 + \left(\frac{\partial^2 F}{\partial z^2} + \frac{\partial^2 F}{\partial x^2} \right) (\frac{\partial F}{\partial y})^2 + \left(\frac{\partial^2 F}{\partial x^2} + \frac{\partial^2 F}{\partial y^2} \right) (\frac{\partial F}{\partial z})^2 \\ -2 \left(\frac{\partial F}{\partial x} \frac{\partial F}{\partial y} \frac{\partial^2 F}{\partial x \partial y} + \frac{\partial F}{\partial y} \frac{\partial F}{\partial z} \frac{\partial^2 F}{\partial y \partial z} + \frac{\partial F}{\partial z} \frac{\partial F}{\partial x} \frac{\partial^2 F}{\partial z \partial x} \right) \end{array} \right]}{\left(\frac{\partial F}{\partial x}^2 + \frac{\partial F}{\partial y}^2 + \frac{\partial F}{\partial z}^2 \right)^{3/2}}$$

Parametric Surfaces: Tangent Plane, Surface Normal, and Curvature

$$\vec{x} = (p(u,v), q(u,v), s(u,v)) \text{ or } x = p(u,v) y = q(u,v) z = s(u,v)$$

Tangent Plane ($\vec{x} = (x,y,z)$, $\vec{\xi} = (\xi, \eta, \zeta)$)

$$(\vec{\xi} - \vec{x}) \cdot (\frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv}) \det \begin{pmatrix} \xi - x & \eta - y & \zeta - z \\ \frac{\partial p}{\partial u} & \frac{\partial q}{\partial u} & \frac{\partial s}{\partial u} \\ \frac{\partial p}{\partial v} & \frac{\partial q}{\partial v} & \frac{\partial s}{\partial v} \end{pmatrix} = 0$$

Normal

$$\frac{\xi - x}{\frac{\partial(q,s)}{\partial(u,v)}} = \frac{\eta - y}{\frac{\partial(s,p)}{\partial(u,v)}} = \frac{\zeta - z}{\frac{\partial(p,q)}{\partial(u,v)}}$$

Mean Curvature

$$\frac{\left(\frac{d\vec{x}}{du} \cdot \frac{d\vec{x}}{du}\right)\left(\frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv} \cdot \frac{d^2\vec{x}}{dv^2}\right) - 2\left(\frac{d\vec{x}}{du} \cdot \frac{d\vec{x}}{dv}\right)\left(\frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv} \cdot \frac{d^2\vec{x}}{dudv}\right) + \left(\frac{d\vec{x}}{dv} \cdot \frac{d\vec{x}}{dv}\right)\left(\frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv} \cdot \frac{d^2\vec{x}}{du^2}\right)}{\left(\frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv} \cdot \frac{d\vec{x}}{du} \times \frac{d\vec{x}}{dv}\right)^{3/2}}$$

3.016 Home

Full Screen

Close

Quit

Graph Surfaces: Tangent Plane, Surface Normal, and Curvature

$$z = f(x, y)$$

Tangent Plane ($\vec{x} = (x, y, z)$, $\vec{\xi} = (\xi, \eta, \zeta)$)

$$\frac{\partial f}{\partial x}(\xi - x) + \frac{\partial f}{\partial y}(\eta - y) = (\zeta - z)$$

Normal

$$\frac{\xi - x}{\frac{\partial f}{\partial x}} = \frac{\eta - y}{\frac{\partial f}{\partial y}} = \frac{\zeta - z}{-1}$$

Mean Curvature

$$\frac{(1 + \frac{\partial f}{\partial x}^2)\frac{\partial^2 f}{\partial y^2} - 2\frac{\partial f}{\partial x}\frac{\partial f}{\partial y}\frac{\partial^2 f}{\partial x \partial y} + (1 + \frac{\partial f}{\partial y}^2)\frac{\partial^2 f}{\partial x^2}}{\sqrt{1 + \frac{\partial f}{\partial x}^2 + \frac{\partial f}{\partial y}^2}}$$

3.016

Representations of Surfaces

notebook (non-evaluated) pdf (evaluated) html (evaluated)

Visualization examples of surfaces represented in three ways: 1) graph $z = f(x, y)$; 2) parametric $(x(u, v), y(u, v), z(u, v))$; 3) level set constant $= F(x, y, z)$.

2: Using `Plot3D` to plot *GraphFunction* .

4: Using `ParametricPlot3D` to visualize a surface of the form $(x(u, v), y(u, v), z(u, v))$ given by *SurfaceParametric* . The lines of constant u and v generate the "square mesh" of the approximation to the surface. Each line on the surface is of the form: $\vec{r_1}(u) = (x(u, v = \text{const}), y(u, v = \text{const}), z(u, v = \text{const}))$ and $\vec{r_2}(v) = (x(u = \text{const}, v), y(u = \text{const}, v), z(u = \text{const}, v))$. The set of all crossing lines $\vec{r_1}(u)$ and $\vec{r_2}(v)$ is the surface. Each little "square" surface patch provides a convenient way to define the local surface normal—because both the vectors $d\vec{r_1}/du$ and $d\vec{r_2}/dv$ are tangent to the surface, their cross-product is either an inward-pointing normal or outward-pointing normal.

8: Using `ContourPlot3D` in the `Graphics‘ContourPlot3D‘` package to visualize the level set formulation of a surface constant $= F(x, y, z)$ given by *ConstFunction* .

12: Animation is produced by using `Table` to generate the level sets for different constants.

Surface of the form: (z=f(x,y))

```
1  GraphFunction[x_, y_] := ((x - y) (x + y))/(1 + (x + y)^2)
2  Plot3D[Evaluate[GraphFunction[x, y]], {x, -1, 1}, {y, -1, 1}]
```

Surface of the form: (x(u,v), y(u,v), z(u,v))

```
3  SurfaceParametric[u_, v_] := {Cos[u] v, u Cos[u + v], Cos[u]}
4  ParametricPlot3D[
     Evaluate[SurfaceParametric[u, v]], {u, -2, 2}, {v, -2, 2}]
5  Table[ParametricPlot3D[
       Evaluate[SurfaceParametric[u, v]], {u, -ep, ep},
       {v, -ep, ep}, PlotRange -> {{-4, 4}, {-4, 4}, {-1, 1}},
       PlotPoints -> {1 + Round[ep/.125], 1 + Round[ep/.125]}],
     {ep, .125, 4.25, .125}]
```

Surface of the form: (F(x,y,z) = constant)

```
6  << Graphics`ContourPlot3D`
7  ConstFunction = x^2 - 4 x y + y^2 + z^2
8  ContourPlot3D[ConstFunction, {x, -1, 1},
     {y, -1, 1}, {z, -1, 1}, Contours -> {6}]
9  cpa = ContourPlot3D[x^2 - 4 x y + y^2 + y x z^2,
       {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours -> {0, 2, 8},
       PlotPoints -> {5, 7}, DisplayFunction -> Identity]
10 cpb = ContourPlot3D[x^2 - 4 x y + y^2 + y x z^2,
       {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
       Contours -> {0, 2, 8}, PlotPoints -> {5, 7},
       ContourStyle -> {{Hue[0]}, {Hue[.25]}, {Hue[.5]}},
       Lighting -> False, DisplayFunction -> Identity]
11 Show[GraphicsArray[{cpa, cpb}]]
12 Table[ContourPlot3D[x^2 - 4 x y + y^2 + y x z^2,
       {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours -> {i},
       PlotPoints -> {5, 7}], {i, -2, 10, .5}]
```

3.016 Home

Full Screen

Close

Quit

Integration over Surfaces

Integration of a function over a surface is a straightforward generalization of $\int\int f(x,y)dxdy = \int f(x,y)dA$. The set of all little rectangles $dxdy$ defines a planar surface. A non-planar surface $\vec{x}(u,v)$ is composed of a set of little parallelogram patches with sides given by the infinitesimal vectors

$$\begin{aligned} \vec{r_u}du &= \frac{\partial \vec{x}}{\partial u}du \\ \vec{r_v}du &= \frac{\partial \vec{x}}{\partial v}dv \end{aligned} \tag{15-5}$$

Because the two vectors $\vec{r_u}$ and $\vec{r_v}$ are not necessarily perpendicular, their cross-product is needed to determine the magnitude of the area in the parallelogram:

$$dA = \|\vec{r_u} \times \vec{r_v}\|dudv \tag{15-6}$$

and the integral of some scalar function, $g(u,v) = g(x(u,v), y(u,v)) = g(\vec{x}(u,v))$, on the surface is

$$\int g(u,v)dA = \int\int g(u,v)\|\vec{r_u} \times \vec{r_v}\|dudv \tag{15-7}$$

However, the operation of taking the norm in the definition of the surface patch dA indicates that some information is getting lost—this is the local normal orientation of the surface. There are two choices for a normal (inward or outward).

When calculating some quantity that does not have vector nature, only the magnitude of the function over the area matters (as in Eq. 15-7). However, when calculating a vector quantity, such as the flow through a surface, or the total force applied to a surface, the surface orientation matters and it makes sense to consider the surface patch as a vector quantity:

$$\begin{aligned} \vec{A}(u,v) &= \|\vec{A}\|\hat{n}(u,v) = A\hat{n}(u,v) \\ d\vec{A} &= \vec{r_u} \times \vec{r_v} \end{aligned} \tag{15-8}$$

where $\hat{n}(u,v)$ is the local surface unit normal at $\vec{x}(u,v)$.

3.016

3.016 Home

Full Screen

Close

Quit

Example of an Integral over a Parametric Surface

notebook (non-evaluated) pdf (evaluated) html (evaluated)

The surface energy of single crystals often depends on the surface orientation. This is especially the case for materials that have covalent and/or ionic bonds. To find the total surface energy of such a single crystal, one has to integrate an *orientation-dependent* surface energy, $\gamma(\hat{n})$, over the surface of a body. This example compares the total energy of such an anisotropic surface energy integrated over a sphere and a cube that enclose the same volume.

1: This is the parametric equation of the sphere in terms of longitude $v \in (0, 2\pi)$ and latitude $u \in (-\pi/2, \pi/2)$.

2: Calculate the tangent plane vectors $\vec{r_u}$ and $\vec{r_v}$

3: Using `CrossProduct` from the `Calculus‘VectorAnalysis‘` package to calculate $\vec{r_u} \times \vec{r_v}$ for subsequent use in the surface integral.

4: Using `DotProduct` to find the magnitude of the local normal.

5: This is the local unit normal $\hat{n}$.

6: This is just an example of a $\gamma(\hat{n})$ that depends on direction that will be used for purposes of illustration.

9: Using `SphericalPlot3D` from the `Graphics‘ParametricPlot3D‘` package to illustrate the form of *SurfaceTension* for the particular choice of $\gamma_{111} = 12$.

10: Using the result from $|\vec{r_u} \times \vec{r_v}|$, the energy of a spherical body of radius $R = 1$ is computed by integrating $\gamma\hat{n}$ over the entire surface.

12: This would be the energy of a cubical body with the same volume.

14: This calculation is not very meaningful, but it is the value of γ_{111} such that the cube and sphere have the same total surface energy. The minimizing shape for a fixed volume is calculated using the *Wulff theorem.*

Example: Integrating an Orientation-Dependent Surface Tension over the surface of a cube and a sphere

```
1   sphereresurf[u_, v_] :=
      R {Cos[v] Cos[u], Cos[v] Sin[u], Sin[v]}

2   Ru[u_, v_] = D[spheresurf[u, v], u] // Simplify
    Rv[u_, v_] = D[spheresurf[u, v], v] // Simplify

3   << Calculus`VectorAnalysis`

4   NormalVector[u_, v_] =
      CrossProduct[Ru[u, v], Rv[u, v]] // Simplify

5   NormalMag =
      Sqrt[DotProduct[NormalVector[u, v], NormalVector[u, v]] //
          Simplify] // PowerExpand

6   UnitNormal[u_, v_] = NormalVector[u, v]/NormalMag

7   SurfaceTension[nvec_] :=
      1 + gamma_111 * nvec[[1]]^2 nvec[[2]]^2 nvec[[3]]^2

8   << Graphics`ParametricPlot3D`

9   SphericalPlot3D[
      SurfaceTension[UnitNormal[u, v]] /. gamma_111 → 12,
      {u, 0, 2 Pi}, {v, -Pi/2, Pi/2}]

10  SphereEnergy = Integrate[
        Integrate[SurfaceTension[UnitNormal[u, v]] Cos[v],
          {u, 0, 2π}], {v, -π/2, π/2}]

11  CubeSide = (4π/3)^(1/3)

12  CubeEnergy = 6 (CubeSide^2 SurfaceTension[{1, 0, 0}])

13  EqualEnergies =
      Solve[CubeEnergy == SphereEnergy, gamma_111] // Flatten

14  N[gamma_111 /. EqualEnergies]
```

3.016 Home

Full Screen

Close

Quit

Index

3.016 Home

Full Screen

Close

Quit

3.016 Home

Full Screen

Close

Quit